

It's a **date**

... says Stephen Wells, who looks at an easy way of arranging dates in spreadsheets, as well as discussing the ins and outs of decimal places.

As a freelance journalist, I don't send out many invoices a month. So it's long been my habit to create invoice numbers by simply repeating the date in reverse. An invoice dated August 31st this year would thus be numbered 950831.

This is not intended to be a secret code; it just avoids maintaining a log of invoice numbers. But it tickled me when one now-defunct magazine used consistently to record on their Remittance Advice slips my invoice date as a month after it actually was. When they quoted my invoice number, they didn't appear to notice that their subterfuge for delaying payment was immediately exposed.

I was reminded of this when I received the latest suggestions from this column's most faithful contributor.

Shane Devenshire, of Walnut Creek, California, has this wonderful ability to fiddle with spreadsheet functions until they produce useful solutions to everyday presentation problems. Although, bless his heart, he tends to start out with Lotus 1-2-3 and guess at the Excel version. So when I make sure everything's correct, I sometimes have to rack my brain.

He says dates are often recorded this way on mainframes and his first suggestion this month is for converting them into the traditional DDMMYY format. (Oddly, he doesn't confuse the issue with the usual American order of MMDDYY. Maybe he's an ex-pat or an anglophile.) If the number were 950831 in cell A21 and you wanted to convert it to 31/8/95, then in Excel in the appropriate cell you'd enter:

```
=DATE(A21/10^4,MOD(A21,10^4)/100,
MOD(A21,100))
```

In Lotus 1-2-3 it would be

```
@DATE(A21/10^4,@MOD(A21,10^4)/100,
@MOD(A21,100))
```

Another conversion which is often needed with many spreadsheet applications is that from decimal parts of an hour to minutes. Let's say that cell A17 reads, in hours, 9.45. The time is not a quarter-to-ten. It's 27 minutes past nine. How can you make it read, 9.27? Shane suggests this formula:

```
=INT(A17)+MOD(A17,1)*0.6
```

The Lotus 1-2-3 equivalent is:

```
@INT(A17)+@MOD(A17,1)*0.6
```

If you don't need to make further calculations from the result, you can use this formula to create a string:

```
=TEXT(INT(A17)+MOD(A17,1)*0.6,
format_text)
```

where format_text is a Name with the definition, = "##.##"

1-2-3 users would enter:

```
@STRING(@INT(A17)+@MOD(A17,1)*0.6,2)
```

To make it a string that reads 9 hours 27 minutes, you could use

```
=TEXT(INT(A17),0)&" hours
"&TEXT(MOD(A17,1)*60,0)&" minutes"
```

In Lotusese, that's:

```
@STRING(@INT(A17),0)&" hours
"&@STRING(@MOD(A17,1)*60,0)&"
minutes"
```

Shane also gives the timely reminder that Excel offers a jolly useful CONVERT function in its Analysis Tool pack. In Excel 4.0 you choose Options, Add-ins, Add, then pick the file, analysis.xla. In Excel 5.0, you choose Tools, Add-ins, and check the box by MS EXCEL 4.0 Analysis Tools.

For the subject in hand, you can convert hours to minutes. If A17 again shows 9.45 in hours, you'd put:

```
=CONVERT(A17,"hr","mn")
```

You'd get the result 567. Of course, if you're making manual entries in cell A17 (I mean, it doesn't contain a formula), in Excel you can enter the real time as 9:27. Excel recognises the colon and in the Formula Bar spells it out as 9:27:00 AM (the Time format) although the cell will just display 9:27 (in the default format). If you want to convert that form into minutes (still 567) you'd use the HOUR and MINUTE functions:

```
=HOUR(A17)*60+MINUTE(A17)
```

Another variation is that Excel will accept the entry 9 space 27/60 and, if the cell is formatted for time, display that in hours as, 9.45. With default formatting it displays logically enough as 9 9/20.

Incidentally, apart from time, the CONVERT function will help you out with all manner of wonderful translations. You can go from teaspoons to fluid ounces, pints to litres, Fahrenheit to Celsius, horsepower to Watts, Joules to calories, meters to miles or inches, and even turn slugs into grams — the slugs not being the garden variety but the unit of mass.

Let's go round again

Finally, Shane has another suggestion about rounding. His last thoughts on this subject I included in the February column.

Rounding calculated currency amounts up or down can often give grief to spreadsheet users, whether in pounds, dollars, or ecus. It is not unusual, in just one invoice, for one rule to apply for, say, a trade discount, and another for VAT. And it can obviously make a difference whether each item is rounded, or just the total of a number of items.

Spreadsheet users frequently run into two problems: what is the correct method to use? And how do you make all the different financial statements balance?

I always follow three little rules:

- Be consistent. Establish a policy and stick to it.
- On the worksheets (even if not on the printed reports) always include a footnote briefly detailing the methodology used.
- Include in the worksheet (not in the printed reports) alternate ways of arriving at disputable totals as a cross-check.

But to get down to Shane's latest tip. Supposing you have a column of figures with odd decimal amounts. In the screenshot *(above right)*, you can see that I've created them in the range A25:A35 with a simple formula.

If you wanted to total them as they are, you could just enter =SUM(A25:A35) in cell A36. But suppose you wanted to round each individual entry to two decimal places. I've done that in column B with the simple formula, =ROUND(A25,2) etc. Totalling the new figures gives a slightly different result. Neither answer is



Rounding decimal figure entries individually, with results on the left and formulas on the right

necessarily wrong. It just depends on the needs of the moment.

But Shane points out that taking another column just for rounding the entries individually can play havoc with your worksheet design, even if you hide column B. So he suggests using an Excel feature I've mentioned before: the Array. In the cell of your choice, you can just enter =SUM(ROUND(A25:A35,2)). To make it an Array formula, you hold down the Ctrl+Shift keys while you press Enter (Command+Enter on the Mac). Excel itself will run down through the range A25:A35, round each separate figure, then total the result. So, as you can see in the illustration, you get the same answer in cell B37 (where the Array formula is) as B36 (the sum of the individually rounded figures).

There's one other handy little Excel hint that can save book-keepers millions of milliseconds (which could add up to a coffee break). Instead of having to keep pressing a full stop key on the regular keyboard or the number keypad, Excel will enter decimal points for you automatically.

Just choose Options, Workspace and the Fixed Decimal check box. If the Status Bar is active (select Options, Workspace, Display Status Bar to turn it on), then the word FIX will appear in the far-right box in the bottom right-hand corner of the screen.

You can override the automatic decimal just by inserting a decimal point on any entry.

How to format

When setting up a new Excel worksheet, I always find it useful to have both the standard toolbar and Formatting toolbar showing. I also take advantage of the many artistically designed table formats which Excel offers.

To display the Formatting toolbar, with the cursor on the toolbar, just press the right mouse button, leave the Standard toolbar checked, but also check the Formatting option.

To access the 14 built-in tabular formats in Excel 4.0, highlight the area of your worksheet to format, choose Format, and AutoFormat, and you arrive at the AutoFormat dialogue box. Having selected a format, you can edit out parts of it if you wish by just choosing Options.

Once back on your worksheet, you can apply that same format to other areas of your worksheet just by highlighting them and choosing the last tool in the Formatting toolbar: the AutoFormat tool. If you press Shift as you click on it, Excel will cycle through the last tabular formats you've used.

Incidentally, if you ever want to eliminate any borders on all or part of your worksheet, just highlight the area and press Ctrl+Shift+ - (hyphen).

Financial analysis

This month we can start to create the tabular and charted results of the 19 ratios in the financial analysis template for service companies — those which don't carry stock.

Those in the first group are called Liquidity Ratios, and in most industries they are all of the Times type (i.e. not a percentage, or in days, or some other form). *Fig 2* gives a listing of the required formulae for the essential cells in rows 34 to 38.

Fig 2 Financial analysis template listing

	A	B
34	LIQUIDITY RATIOS	(TIMES)
35		=$B$1
36	Current	=Current_Assets/Current_Liabilities
37	Quick	=Quick_Current_Assets/Current_Liabilities
38	Current Liabs to Net Worth	=Current_Liabilities/Net_Worth

Because this template uses Names throughout, the listing needs to show only column B. On the screen, exactly the same formulae are shown in columns C, D, E and F. Or they can be if you created the Names as explained in July's column.

The only exception is the date row which has =$B$1 in cell B35, then =B35-1 in C35, =C35-1 in D35, and so on. Rows 33 and 39 are blank, just to space out the groups.

Column G holds the average results for the company's industry. These ratios may be available from a commercial source like Dun and Bradstreet, or from a trade association, or the DTI (Dept of Trade & Industry). In the example given, the business is an advertising agency.

If you entered the sample figures for the Balance Sheet, given in July's column, and for the Income Statement, illustrated in last month's column, then you will see the results shown in *Fig 3*.

An Annual Income Statement shows what has happened to a company during a year. A Balance Sheet is a snapshot of the state of the business on the last day of a fiscal period (quarter or year). But reviewing a table of financial ratios for the past five years is like looking at a moving picture of the activities of the company.

Creating charts

You can compare the results with others in the company's industry. But more importantly, you can catch trends which may need management's attention. Trends are spotted most easily with a chart. The template, written on Excel 4.0 but importable to other spreadsheets, is set up to create them easily. Excel offers a number of ways of making charts, but here's a quick way, using the mouse.

Open the worksheet we've been creating, SERVICE.XLS. Select the range, A35:G38. Select, Edit, Copy. Then select, File, New, Chart, followed by Edit, Paste Special. In the dialogue box choose Values in Columns, Series Names in First Row, and Categories in First Column.

Then select, Chart, Add Legend, and then, Chart, Attach Text and choose Chart Title. In the Formula Bar, enter an equals sign and select Window, SERVICE.XLS, cell A34. Delete the word "Title" on the end of the line.

That's it. You should now have the chart shown in *Fig 4*. You can see right away that this company is maintaining stable liquidity. The only thin end of a trend is the growth in Current Liabilities to Net Worth in '93 and '94. But is the company in a good position? Let's take a moment to examine these measurements of solvency.

A high ratio means higher than the industry or growing for the company. A high Current Ratio maintained throughout the year usually indicates a company has adequate clients' accounts coming due to enable it to pay the amounts outstanding to suppliers in the same period, after the company's cash and sales income during the following weeks are included.

A low Current Ratio might indicate an inability to meet current obligations, but it can also be a sign of extraordinary managerial ability. Average interest rates reported in the news can be deceptive. Diligent financial officers will compare one-year and 30-day rates. Current liabilities can be a cheap source of funds when interest rates on short-term loans are below rates on long-term debt.

To summarise: the higher the Current Ratio, the better the indication of solvency, but the quality and management of assets must also be considered.

A high Quick Ratio implies that the company can meet unexpected demands for working capital. If it fell below one, the company would probably have to incur or increase long-term debt. In the example given, the company ratio is consistently above one and in line with its industry.

To summarise: the Quick Ratio differs from the Current Ratio in that the values of pre-paid expenses and other current assets are excluded. It measures quick assets, those that are highly liquid, meaning immediately convertible to cash.

The higher the Current Liabilities to Net Worth ratio is, the less security there is for the creditors of the company. In the example, the figure is not only higher than the average for the industry, but more worryingly, it is growing. From the stockholders' viewpoint, the more the suppliers' funds can be used instead of their own, the more beneficial it may appear to be. But when suppliers are pressing for payment, management attention is diverted from the business.

To summarise: this ratio contrasts the funds that creditors are risking against the funds invested by the stockholders or owners. It compares what's owed in the current year with what's owned.

Next month we'll look at some of the activity and gearing ratios.

Microsoft Excel

File Edit Formula Format Data Options Macro Window Help

Normal

F38 =Current_Liabilities/Net_Worth

	A	B	C	D	E	F	G
33							
34	LIQUIDITY RATIOS (TIMES)						
35		1994	1993	1992	1991	1990	INDUST
36	Current	1.14	1.18	1.31	1.24	1.20	1.20
37	Quick	1.14	1.12	1.24	1.24	1.09	1.10
38	Current Liabilities to Net Worth	3.55	2.65	1.72	2.03	2.19	1.96
39							



Fig 3 (above) *Example results for the Liquidity Ratios portion of the financial analysis template for service companies*

Fig 4 (left) *The liquidity position of the example results in chart form*

PCW Contacts

Stephen Wells welcomes feedback on spreadsheets via *PCW* Editorial at the usual address, or at **stephen_wells@pcw.ccmail.compuserve.com**

For the financial analysis Excel templates for service companies and those which carry stock, send a formatted 3.5in disk and a stamped, self-addressed envelope.